

Anything Anywhere All At Once: Output Graphs, Tables, and Text As Any Composite You Wish

LeRoy Bessler

Le_Roy_Bessler@wi.rr.com

Bessler Consulting and Research

Visual Data Insights™

Strong Smart Systems™

Mequon, Wisconsin, USA

Tutorial based on Visual Data Insights Using SAS ODS Graphics: A Guide to Communication-Effective Data Visualization

Chapter 14

“Delivering Precise Numbers and Alternative Views for Graphs Using SAS ODS HTML5”

Other web pages, but also includes creating composites for web pages

ODS LAYOUT ABSOLUTE

used in Chapter 13

can put

Anything Anywhere All At Once

It is available only with


ODS PRINTER and ODS PDF

Tutorial
Will Show
You
How To Get
This Result


The 2 BY 2
layout
is NOT
required

Student Average Height & Detail, By Gender

Female Avg Height By Age



Male Avg Height By Age



Female Student Information

Name	Age	Height
Alice	13	56.5
Barbara	13	65.3
Carol	14	62.8
Jane	12	59.8
Janet	15	62.5
Joyce	11	51.3
Judy	14	64.3
Louise	12	56.3
Mary	15	66.5

Male Student Information

Name	Age	Height
Alfred	14	69.0
Henry	14	63.5
James	12	57.3
Jeffrey	13	62.5
John	12	59.0
Philip	16	72.0
Robert	12	64.8
Ronald	15	67.0
Thomas	11	57.5
William	15	66.5

Data Source: [SASHELP.CLASS Data Set](#)

Building the Composite

Tables must be created
as image files
to be programmatically imbedded
at the desired places

NOTE:

**On the ODS PRINTER statement,
DPI=300 is a holdover
from code used
to prepare the book manuscript
for optimal printing results.**

**NOT inherently required:
omitting it causes
fuzzier text, lines, and curves.**

1. Create two tables as image files
2. Build the composite:
 - a. Inlay a common title
 - b. Create two graphs in-place
 - c. Insert two tables from disk
 - d. Inlay a common footnote
3. Put a border around the composite


```
/* The process might require tuning  
to get a satisfactory result.  
Gather controls  
likely to need change  
up here at top of the code. */
```

```
%let GeneralFontSize = 11pt; /* for all text,  
    except any to which override is applied */
```

```
%let TableCellHeight = 8pt;  
/* used for creating tables as images  
with table rows compacted */
```

```
%let WhiteSpace = 0.225in; /* between graphs row  
                             and tables row */  
  
/* title and footnote 13pt Arial Bold */  
%let TitleFootnoteTextAttrs =  
    %str(fontfamily='Arial'  
        font_size=13pt font_weight=Bold);  
  
/* center them and make them black */  
%let CommonTitleAttrs =  
    &TitleFootnoteTextAttrs  
    just=center color=black;  
%let CommonFootnoteAttrs =  
    &TitleFootnoteTextAttrs  
    just=left color=blue;
```

```
%let CommonTitle =  
%str(Student Average Height & Detail, By Gender);  
  
%let CommonFootnote =  
%str(Data Source: SASHELP.CLASS Data Set);
```

```
/* To build tables as images:  
   create a temporary ODS style  
   that is deleted after the build.
```

```
Use PROC TEMPLATE */
```

```
proc template;
```

```
define style ToBeDeleted;
```

```
parent=styles.PRINTER;
```

```
class Fonts /
```

```
  'TitleFont'      =
```

```
    ("Arial", &GeneralFontSize, Bold)
```

```
  'headingFont'    =
```

```
    ("Arial", &GeneralFontSize, Bold)
```

```
  'docFont'        =
```

```
    ("Arial", &GeneralFontSize, Bold)
```

```
;
```

```
class Header      /  
    backgroundcolor = white color = black;
```

```
class RowHeader /  
    backgroundcolor = white color = black;
```

```
class data      /  
    backgroundcolor = white color = black;
```

```
/* Start to build table images */
```

```
/* Defaults are DATE NUMBER CENTER */
```

```
options nodate nonumber;
```

```
options papersize=(2.11in 2.4in);
```

```
/* table image width & height */
```

```
ods results off;
```

```
ods _all_ close;
```

```
ods printer style=ToBeDeleted dpi=300
  printer=PNG file="C:\temp\RightTable.png";

title1 justify=center color=black
  "Male Student Information";
proc print data=SASHELP.CLASS noobs;
where Sex EQ 'M';
id Name;
var Age Height;
run;
```



```
ods printer style=ToBeDeleted dpi=300  
  printer=PNG file="C:\temp\LeftTable.png";
```

```
title1 justify=center color=black  
  "Female Student Information";
```

```
proc print data=SASHELP.CLASS noobs;  
where Sex EQ 'F';  
id Name;  
var Age Height;  
run;
```

```
title1; /* Prevent TITLE1 being inherited.  
  In THIS example, it would cause  
  "Female Student Information"  
  to also appear at the TOP of the composite. */
```

```
/* End building tables as images  
by closing the ODS PRINTER destination  
and deleting the temporary style. */
```

```
ods printer close;
```

```
/* The style used is specific to this experiment.  
I am deleting it. */
```

```
proc template;  
delete ToBeDeleted /  
store=sasuser.templat;  
run;
```

```
/* Now, start to build the composite */
```

```
/* Defaults are DATE NUMBER CENTER */
```

```
options nodate nonumber;
```

```
options papersize=("5in" "5.69in");
```

```
/* composite width & height */
```

```
ods results off;
```

```
ods _all_ close;
```

```
ods printer printer=PNG dpi=300
  /* default style for ODS PRINTER accepted */
  file="C:\temp\Fig-3_Composite.png";

/* ODS PRINTER creates the composite image */

/* ODS LAYOUT places the correct items
   in the desired places */

ods layout absolute;

/* Location and size of desired placement
   is defined by REGION statements. */
```

```
/* Inlay a common title */  
  
ods region x=0in y=0in  
  /* These x & y values are at the upper left  
    corner of the "paper".  
    They define the location  
    of the upper left corner of the region. */  
width=5in height=0.6in;  
/* width is measured right from the left */  
/* height is measured down from the top */
```

```
/* Use PROC ODSTEXT for all text parts  
not created inside the images. */
```

```
proc odstext;
```

```
/* First, some white space (if needed): */
```

```
p ' ' / style={font_size=0.1in};
```

```
/* Now the title: */
```

```
p "&CommonTitle" / style={&CommonTitleAttrs};
```

```
run;
```

```
/* White space can provide a finer increment  
than the effect of  
changing Y= on the REGION statement */
```

```
/* setup for BOTH graphs: */  
ods graphics on / reset=all  
    /* override default SCALE=ON */  
    scale=off width=2.5in height=2in noborder;  
                                /* default is BORDER */  
  
/* define the region for the left graph */  
ods region x=0in y=0.6in  
/* 0.6 inches down from the upper left corner  
   of the composite image */  
    width=2.5in height=2in;  
  
ods graphics / reset=index  
    /* RESET=INDEX means inherit all prior  
       ODS GRAPHICS statement specifications,  
       but allow IMAGENAME to be defined here. */  
    imagename="LeftGraph";
```

```
title justify=left
  font=Arial Bold height=&GeneralFontSize
  'Female Avg Height By Age';
proc sgplot data=sashelp.class noborder;
where Sex='F';
vbar age / response=height stat=mean
  displaybaseline=off
  datalabel
  datalabelattrs=(family=Arial
    size=&GeneralFontSize weight=Bold)
  nooutline
  fillattrs=(color=red)
  barwidth=0.5;
yaxis display=none;
xaxis display=(nolabel noline noticks)
  valueattrs=(family=Arial
    size=&GeneralFontSize weight=Bold);
format age 2. height 2.; run;
```



```
/* same code as for LeftGraph
   except for these changes in RED: */
ods region x=2.5in y=0.6in
   width=2.5in height=2in;
ods graphics / reset=index
   imagename="RightGraph";
title justify=left
   font=Arial Bold height=&GeneralFontSize
   'Male Avg Height By Age';
proc sgplot data=sashelp.class noborder;
where Sex='M';
/* same PROC SGPLOT step end code here:
   VBAR, YAXIS, XAXIS, FORMAT statements, RUN */
```

```
/* insert the RightTable image
   below white space */

ods region
  x=2.5in y=2.6in /* at the bottom of the region
                   allocated to right graph */
  width=2.5in height=2.84in;

proc odstext;
p ' ' / style={font_size=&WhiteSpace};
run;

data _null_;
dcl odsout obj();
obj.image(file: "C:\temp\RightTable.png");
run;
```

```
/* Inlay a common footnote. */

ods region
  x=0in y=5.44in /* at the bottom of the regions
                  allocated to the graphs */
  width=5in height=0.25in;
/* the bottom of this region is at 5.69in which
   corresponds to the height of the composite */

proc odstext;
p "    &CommonFootnote" /
  style={asis=on &CommonFootnoteAttrs};
/* ASIS=ON preserves
   the three leading blanks for an indent */
run;
```

```
ods layout end;
```

```
ods printer close; /* Finished */
```

```
/* Fig-3_Composite.png, LeftTable.png, and  
RightTable.png are in C:\temp.  
The table images can be deleted or ignored. */
```

```
/* NEXT: Use a macro  
to apply a border to the composite.  
This border is optional. */
```

```
/* About Borders:
```

Borders on graphs were suppressed, using
NOBORDER option on ODS GRAPHICS statement.


Borders on the tables as images
COULD BE applied,
but it requires the same post-processing
to be used here for the composite. */

```
/* Use macro to apply border to the composite */  
/* This border is optional. */  
  
/* Source code for ApplyBorderToAnyImage.sas  
   is in the paper. Store it wherever you like */  
  
%include  
    "YourFolderLocation\ApplyBorderToAnyImage.sas";  
%ApplyBorderToAnyImage(  
in=C:\temp\Fig-3_Composite.png,  
out=C:\temp\Fig-3_CompositeWithBorder.png,  
width=5in,  
height=5.69in);  
  
/* Now, Fig-3_Composite.png  
   can be deleted or ignored. */
```


The Result

Student Average Height & Detail, By Gender

Female Avg Height By Age



Male Avg Height By Age



Female Student Information

Name	Age	Height
Alice	13	56.5
Barbara	13	65.3
Carol	14	62.8
Jane	12	59.8
Janet	15	62.5
Joyce	11	51.3
Judy	14	64.3
Louise	12	56.3
Mary	15	66.5

Male Student Information

Name	Age	Height
Alfred	14	69.0
Henry	14	63.5
James	12	57.3
Jeffrey	13	62.5
John	12	59.0
Philip	16	72.0
Robert	12	64.8
Ronald	15	67.0
Thomas	11	57.5
William	15	66.5

The book's Chapter 14

“Delivering Precise Numbers and
Alternative Views for Graphs Using
SAS ODS HTML5”

includes several ways to create
composites for web pages

ODS LAYOUT GRIDDED
is used in Chapter 14
with ODS HTML5.

That coding is simpler.

ODS LAYOUT ABSOLUTE

is also used in Chapter 14
with ODS HTML5.

First, a standalone image is
created for the composite,
using ODS PRINTER.

Then it is programmatically
imbedded in the web page.

See the details in the book.

Limitation

When an ODS LAYOUT
ABSOLUTE image is imbedded
in an ODS HTML5 web page,
NO data tips (mouseover text)
can be provided.

Figure 10. Graphs over table, using common title, common footnote, gridded layout





Figure 11. Graphs over tables, common title, common footnote, in a border, absolute layout



Your comments and questions
are welcome. Contact me at:

Le_Roy_Bessler@wi.rr.com

LeRoy Bessler PhD

Bessler Consulting and Research

Visual Data Insights™

Strong Smart Systems™

Mequon, Wisconsin, USA